
C4µC  -  ADC

● An Analog-to-Digital Converter is a hardware peripheral 
that can convert (by sampling) an analog value (usually 
expressed in Volts/milliVolts) to a digital (binary) value

● Example:    2.5 V --> 10000101

● The parameters to consider when using an ADC are:

● Bit resolution (influences the max number of possible resulting 
values)

● Accuracy (minimum appreciable difference in the analog) signal

● Conversion time (time needed to output a stable value) and the 
speed (expressed in sample/s)

● Number of channels

● Input range (range of accepted inputs, ex. 0-5 V)



  

C4µC  -  ADC

● On the ATMega328p we have (datasheet p.305) an 
ADC with:

● 10 bit resolution (1024 possible values)
● A maximum speed of ~77000 samples/s
● 30-260 us conversion time
● 6 channels
● 0 – Vcc input range (0 -> 3.3V or 0 -> 5V) with a selectable 

1.1V voltage internal reference
● Free-running conversions or single conversion modes



  

C4µC  -  ADC

● How to use the ADC

● With the Arduino API:

– result = analogRead(pin)

– where pin is an Ax pin (ex. A0)

● Manually by manipulating registers:

– First, start the conversion by unsetting the PRR register (bit 
PRADC) and setting the ADCSRA register (bit ADSC).

– During the conversionthe ADCS bit stays HIGH. 
– When it goes LOW, the conversion is finished and the 

retrieved value can be read from the two registers ACDL (low 
8 bits) and ACDH (hight 2 bits)



  

C4µC  -  ADC

● sketch_c4uc_8.1_ADC_AAPI
int analogPin = 3;
int val = 0;

void setup(){
  Serial.begin(115200);
  while(!Serial);
}

void loop(){
  val = analogRead(analogPin);
  Serial.println(val);
}

● Let's inspect the API function:

● analogRead(pin)

– declared in Arduino.h
– defined in wiring_analog.c

Code:
Step 1: retrieve the correct ADC 
channel from the pin number
Step 2: set the MUX properly
Step 3: start the conversion
Step 4: wait until completion
Step 5: retrieve the result



  

C4µC  -  ADC

● Re-organize in: sketch_c4uc_8.2_ADC
#include <avr/io.h>
#include <avr/delay.h>

#define ADC_PIN 0          // ADC 0
#define ADC_THRESHOLD 512  // Our threshold for turning the LED ON or OFF

int main(void){
  Serial.begin(9600);      // Today we are lazy... :-)
  while (!Serial);
  
  DDRB  |= _BV(PORTB5);
  ADCSRA |= _BV(ADEN);     // Enable the ADC, see datasheet section 28.9.2

  // Set reference voltage to Vcc (REFSn=01), see datasheet section 28.9.1, table 28-3
  ADMUX |= (1<<REFS0);

  // Set the lower 3 bits of the MUX register to the pin number, 
  // see DS section 28.9.1, table 28-4
  ADMUX &= 0xF0;
  ADMUX |= ADC_PIN;

  // Set freq prescaler to 128 
  // See datasheet section 28.9.1, table 28-5 
  ADCSRA |= 0b00000111; // Or, in other words.. (1<<ADPS2)|(1<<ADPS1)|(1<<ADPS0) 



  

C4µC  -  ADC

● Re-organize in: sketch_c4uc_8.2_ADC
  while (1) {
    uint16_t value = 0;

    ADCSRA |= _BV(ADSC);             // Start conversion
    
    while ( (ADCSRA & _BV(ADSC)) );  // wait until completion

    // we could recreate the 10-bit value...
    //     value = ((ADCH & 0b00000011) << 8) | ADCL0;
    // ...but we can use the ADC macro 
    value = ADC;
    
    Serial.print(value);
    Serial.print('\n');
    
    if (value > ADC_THRESHOLD)
      PORTB |= _BV(PORTB5);
    else
      PORTB &= ~_BV(PORTB5);

    _delay_ms(1000);
  }
}



  

C4µC  -  ADC

Questions??



  

C4µC  -  PWM

● The Pulse-Width Modulation (PWM) is channel 
modulation in which data is expressed by the duty-
cycle of a square wave

● Useful to produce a 
power-equivalent DC voltage

● Example:

● 3V, 6V or 10.8V
cycling on 12V



  

C4µC  -  PWM

● Manipulating the duty-cycle, it is possible to have a 
square signal that have the same “area” of another 
(non-square) signal



  

C4µC  -  PWM

● PWM is useful for:

● Power-control small peripherals

● LEDs' brightness (this is how your laptop backlight works)

● DC (servo) motors (PWM vs variable DC voltage)

● PWM can be implemented "manually" via delay()
// Example AAPI (bit-banging)
void setup() {
      pinMode(13, OUTPUT);
}
void loop() {
      digitalWrite(13, HIGH);
      delay(100);
      digitalWrite(13, LOW);
      delay(200);
      digitalWrite(13, HIGH);
      delay(400);
      digitalWrite(13, LOW);
      delay(250);
      digitalWrite(13, HIGH);
      delay(150);
      digitalWrite(13, LOW);
      delay(350);
      /* .... */ 
}

// Example AVR (bit-banging) low level
#include <avr/io.h>
#include <util/delay.h>
void main(void) {
      DDRB |= _BV(DDRB5);
      PORTB |= _BV(PORTB5);
      _delay_ms(100);
      PORTB &= ~_BV(PORTB5);
      _delay_ms(200);
      PORTB |= _BV(PORTB5);
      _delay_ms(400);
      PORTB &= ~_BV(PORTB5);
      _delay_ms(250);
      PORTB |= _BV(PORTB5);
      _delay_ms(150);
      PORTB &= ~_BV(PORTB5);
      _delay_ms(350);
      /* .... */ 
}



  

C4µC  -  PWM

● The timers on the ATMega328p have PWM support 
(datasheet Ch. 19, 20)

● Fast PWM

– "The Fast Pulse Width Modulation or Fast PWM modes 
provide ahigh frequency PWM waveform generation 
option"

● Phase-correct PWM

– "The Phase Correct PWM mode provides a high 
resolution, phase correct PWM waveform generation"

● Phase and Frequency Correct PWM

– "The phase and frequency correct Pulse Width 
Modulation, or phase and frequency correct PWM mode 
provides a high resolution phase and frequency correct 
PWM waveform generation option"



  

C4µC  -  PWM

● Fast PWM

● A timer (ex. TC2) is used as a free-running counter. 
When enabled, the Fast PWM uses the compare registers to 
set the PWM signal according the configuration

– Example: start HIGH, set LOW when count==63 (25% 
duty cycle)

● In order to configure a timer to use Fast PWM we need to 
set:

– TCCR2A
– TCCR2B

● Check datasheet for more information (p.203)



  

C4µC  -  PWM

● Fast PWM

● Example:

#include <avr/io.h>
#include <util/delay.h>

void main(void){

  DDRD |= _BV(DDRD3);
  TCCR2A |= 0b10100011;
  TCCR2B |= 0b00001001;
  OCR2A = 63;
  OCR2B = 0;
  while(1)
    _delay_ms(1000);

}

NOTE: code not checked



  

C4µC  -  PWM

● Fast PWM: sketch_c4uc_8.3_fastPWM_motor
#define MAX_VAL 160 // Freq about 100KHz
uint8_t value = 20;
void setup(){
  Serial.begin(9600);
  while (!Serial);
  DDRD |= _BV(PORTD3);
  TCCR2A |= 0b00100011;
  TCCR2B |= 0b00001001;
  OCR2A = MAX_VAL ;
  OCR2B = 0; // value 
}

void loop(){
  while (!Serial.available());
  switch (Serial.read()) {
    case '+':
      value += 10; 
      if(value > MAX_VAL) value = MAX_VAL;
      break;
    case '-':
      value -= 10;
      if(value < 0) value = 0;
      break;
    case 'r':
      value = MAX_VAL >> 3;
  }
  OCR2A = (uint8_t)value;
  delay(50);
}

NOTE: we need an external  
chip to drive the motor



  

C4µC  -  PWM

Questions??



  

C4µC  -  GPIO bit-banging

● In some situations we need to implement a non-
standard communication protocol OR we do not have 
enough specialized ports needed by the application

● Example: two UARTs

● One way to solve the problem is to use the GPIO ports 
manually (bit-banging)

● We manipulate GPIO ports directly, simulating the target 
protocol



  

C4µC  -  GPIO bit-banging

● Morse code is a method of transmitting text information as a 
series of on-off tones, lights, or clicks that can be directly 
understood by a skilled listener or observer without special 
equipment... (Wikipedia)

● We can implement the Morse code and drive the onboard or an 
external LED



  

C4µC  -  GPIO bit-banging

● Example: sketch_c4uc_8.4_bitbanging_Morse

char stringToMorseCode[] = "sos morse code test";

int PIN = 13;

int dotLen = 100;     // length of the morse code 'dot'
int dashLen = dotLen * 3;    // length of the morse code 'dash'
int elemPause = dotLen;  // length of the pause between elements of a character
int Spaces = dotLen * 3;     // length of the spaces between characters
int wordPause = dotLen * 7;  // length of the pause between words

void setup()
{                
  pinMode(PIN, OUTPUT); 
}

void loop()
{ 
  for (int i = 0; i < sizeof(stringToMorseCode) - 1; i++)
    GetChar(stringToMorseCode[i]);
  LightsOff(8000);      
}



  

C4µC  -  GPIO bit-banging

● Example: sketch_c4uc_8.4_bitbanging_Morse

// DOT
void MorseDot()
{
  digitalWrite(PIN, HIGH);    // turn the LED on 
  delay(dotLen);              // hold in this position
}

// DASH
void MorseDash()
{
  digitalWrite(PIN, HIGH);    // turn the LED on 
  delay(dashLen);             // hold in this position
}

// Turn Off
void LightsOff(int delayTime)
{
  digitalWrite(PIN, LOW);     // turn the LED off   
  delay(delayTime);             // hold in this position
}



  

C4µC  -  GPIO bit-banging

● Example: sketch_c4uc_8.4_bitbanging_Morse

void GetChar(char tmpChar)
{
  switch (tmpChar) {
    case 'a': 
      MorseDot();
      LightsOff(elemPause);
      MorseDash();
      LightsOff(elemPause);
      break;
    case 'b':
      MorseDash();
      LightsOff(elemPause);
      MorseDot();
      LightsOff(elemPause);
      MorseDot();
      LightsOff(elemPause);
      MorseDot();
      LightsOff(elemPause);
      break;
    case 'c':
//....

Exercise: make a smarter implementation...



  

C4µC  -  GPIO bit-banging

● When we have to use special hw devices such as:

● Led driver WS2811

– Driver 3 channels
– Data In, Data-out

● Led "embedded driver" WS2812

– Driver and LEDS in single package



  

C4µC  -  GPIO bit-banging

● These drivers can be connected in chains
● DOUT of a driver is connected to next driver's DIN
● A non standard serial communication protocol is used to

– Bit time is 1.25 µs; frequency is 800 kHz
– Different HIGH and LOW times for coding "0" or "1"
– A long LOW is Reset 



  

C4µC  -  GPIO bit-banging

● Data (bitstream) is pushed in the 1st drivers DIN
● Each color is coded with 8 bit (255 levels + off)
● RGB (or in specific case GRB) so 24 bit (3bytes) per driver

– Ex. 4 drivers in chain --> 3x4 = 12 bytes
● A Reset command means "apply pattern"

● Each driver keeps its bytes
and forwards other's



  

C4µC  -  GPIO bit-banging

● We have 2 LED strips (of 8 in row)

● There exists (at least) a library for Arduino:

– Adafruit Neopixel
● Copy and extract in your path:

– ...Arduino\Sketches\libraries\Adafruit_NeoPixel

– Maybe we have to restart the IDE

● See the modified example from the library

– sketch_c4uc_8.5_NeoPixel_uart-cycler

– If we send chars ("0" to "9") with Serial monitor
● we have the applcation of different patterns

and animations

See code & comments



  

C4µC  -  GPIO bit-banging

● Adafruit_NeoPixel library:
● has a class for the management of the "pixels" (WS2812)
● has bit-banging implementation for the protocol

– for different platforms (µC)
– for different frequencies
– at low-level (see assembly in the lib source code)

● An extraction from the library is shown in the next example



  

C4µC  -  GPIO bit-banging

● sketch_c4uc_8.6_bitbanging_asm

#define NUM_PIXELS 16

const uint16_t numBytes = NUM_PIXELS * 3;
uint8_t pixels[numBytes] = { 0 };  // GRB
uint32_t endTime;       // Latch timing reference
volatile uint8_t *port;         // Output PORT register
uint8_t pinMask;

inline bool canShow(void) { return (micros() - endTime) >= 50L; }

void setup() {
  // 
  DDRD |= _BV(DDD7);        // data on pin 7 PB7
  PORTD &= ~_BV(PORTD7);    // set LOW
  port = &PORTD;
  pinMask = _BV(PORTD7);
  
  pixels[0] = 0xEE;         // we set only few of the 16 x 3 bytes
  pixels[4] = 0x8E;
  pixels[7] = 0x8E;
  pixels[8] = 0x2E;
  pixels[47] = 0xff;
}

void loop() {
  show();
  delay(1000);
}



  

C4µC  -  GPIO bit-banging

● sketch_c4uc_8.6_bitbanging_asm

void show(void){
  while(!canShow());
  volatile uint16_t
    i   = numBytes; // Loop counter
  volatile uint8_t
   *ptr = pixels,   // Pointer to next byte
    b   = *ptr++,   // Current byte value
    hi,             // PORT w/output bit set high
    lo;             // PORT w/output bit set low

    // WS2811 and WS2812 have different hi/lo duty cycles; this is
    // similar but NOT an exact copy of the prior 400-on-8 code.

    // 20 inst. clocks per bit: HHHHHxxxxxxxxLLLLLLL
    // ST instructions:         ^   ^        ^       (T=0,5,13)

    volatile uint8_t next, bit;

    hi   = *port |  pinMask;
    lo   = *port & ~pinMask;
    next = lo;
    bit  = 8;
    //...

Let's briefly analise the asm code...



  

C4µC  -  GPIO bit-banging

● sketch_c4uc_8.6_bitbanging_asm

    asm volatile(
     "head20:"                   "\n\t" // Clk  Pseudocode    (T =  0)
      "st   %a[port],  %[hi]"    "\n\t" // 2    PORT = hi     (T =  2)
      "sbrc %[byte],  7"         "\n\t" // 1-2  if(b & 128)
       "mov  %[next], %[hi]"     "\n\t" // 0-1   next = hi    (T =  4)
      "dec  %[bit]"              "\n\t" // 1    bit--         (T =  5)
      "st   %a[port],  %[next]"  "\n\t" // 2    PORT = next   (T =  7)
      "mov  %[next] ,  %[lo]"    "\n\t" // 1    next = lo     (T =  8)
      "breq nextbyte20"          "\n\t" // 1-2  if(bit == 0) (from dec above)
      "rol  %[byte]"             "\n\t" // 1    b <<= 1       (T = 10)
      "rjmp .+0"                 "\n\t" // 2    nop nop       (T = 12)
      "nop"                      "\n\t" // 1    nop           (T = 13)
      "st   %a[port],  %[lo]"    "\n\t" // 2    PORT = lo     (T = 15)
      "nop"                      "\n\t" // 1    nop           (T = 16)
      "rjmp .+0"                 "\n\t" // 2    nop nop       (T = 18)
      "rjmp head20"              "\n\t" // 2    -> head20 (next bit out)
     "nextbyte20:"               "\n\t" //                    (T = 10)
      "ldi  %[bit]  ,  8"        "\n\t" // 1    bit = 8       (T = 11)
      "ld   %[byte] ,  %a[ptr]+" "\n\t" // 2    b = *ptr++    (T = 13)
      "st   %a[port], %[lo]"     "\n\t" // 2    PORT = lo     (T = 15)
      "nop"                      "\n\t" // 1    nop           (T = 16)
      "sbiw %[count], 1"         "\n\t" // 2    i--           (T = 18)
       "brne head20"             "\n"   // 2    if(i != 0) -> (next byte)
      : [port]  "+e" (port),
        [byte]  "+r" (b),
        [bit]   "+r" (bit),
        [next]  "+r" (next),
        [count] "+w" (i)
      : [ptr]    "e" (ptr),
        [hi]     "r" (hi),
        [lo]     "r" (lo));

  endTime = micros(); // Save EOD time for latch on next call
} Asm is not argument for the exam,

but here is an interesting example of what can be done



  

C4µC  -  GPIO bit-banging

● What can be done with LEDs and LED matrixes?

● These panels are complex systems,

● Embedded Systems!

● In the pics an array of 10 x 4 panels

● Each panel has 42 x 42 leds (1764 pixels)

● Not sure if it is WS2812...
... but you have an idea of what
is going on inside



  

C4µC  -  GPIO bit-banging

Questions??

