
Exercise 1

● A bluetooth module need to be connected to our Arduino 
Nano board:

● The module should be powered (and switched ON/OFF) from the 
board itself

● For that reason it will be connected to GPIO pins and, since each 
pin can source a limited amount of current, 3 GPIO pins will be 
used.

● Challenge: the 3 GPIO should be driven toghether and 
instantaneously

● Choose 3 pins in the set of available GPIOs, sketch the connection 
diagram and write 3 functions to control: the initialization of the GPIO, 
the BT power ON, the BT power OFF; .

● void init_bt_pins(); //to be called once in the "setup code"

● void bt_power_on();  //to be called during the application exec

● void bt_power_off(); //to be called during the application exec



  

Exercise 2

● The bluetooth module, previously connected to Arduino 
Nano board for the power supply, is connected also to the 
UART tx-rx pins:

● Consider that the pins are shared with the USB-UART converter 
onboard, see the scheme below; for that reason, the use of the 
UART for debug messages should be limited

● The µC needs to send a sequence of special messages (AT 
commands) to the module (serial connection 9600 8N1):

– ..., "AT+RESET", "AT+IMME1", ..., "AT+DISI?"

Exercise 2

● To each command corresponds an answer that can arrive, from 
the module, within a certain time (variable, unpredictable, with 
other data in between)

● Before sending a command, the answer to the previous command 
must be received or timed out (error signaled)

● The complete table of commands and answers is:

Seq. Command Answer

1 AT+RENEW OK+RENEW

2 AT+RESET OK+RESET

3 AT+IMME1 OK+Set:1

4 AT+ROLE1 OK+Set:1

5 AT+DISI? … OK+DISCE

6 AT+START OK+START



  

Exercise 2

● Propose at least 2 possible strategies to read the answer 
and decide if it is possible to send the next command
(write a flow-chart or functions in pseudo-code)

● Implement the following function, that reads for a maximum time of 
millis_timeout from the serial, comparing the read chars with the 
String resp passed as parameter; can use a parameter dbg for test 
purposes

● // the function returns 0 if no response received within
// millis_timeout
// or returns 1 if a sequence matching resp is found in the stream
uint8_t wait_for_resp(String resp, uint16_t millis_timeout, bool 
dbg)

● /* the function will be used in the sequence like:
  Serial.print("AT+RENEW");
  r = wait_for_resp("OK+RENEW", 100, 0); 
*/

● // to keep the time, in the function implementation, you can 
// use variables and millis() like:
unsigned long int quit_time = millis() + ...



  

Solution 1

● Go back to lesson on Comm.Interfaces and see that in 
"sketch_c4uc_5.1_parallel_sender" we have already used the direct 
access to GPIO registers to control pins together... 

● Three pins could be from PORT B: PB0, PB1, PB2 (Ard. 8, 9, 10)

● The init and control functions:
● #define BT_POWER_MASK 0x07

● void init_bt_ports(){
  DDRB |= BT_POWER_MASK;
  bt_off();
}

● void bt_on(){
  PORTB |= BT_POWER_MASK;
}

● void bt_off(){
  PORTB &= ~BT_POWER_MASK;
}



  

Solution 2

● One simple possible strategy is, for each command to send:

● Send the command 

● Wait a fixed amount of time

● Read and compare the response (to the expected one), if any

● Another is:

● Send the command 

● Wait a certain number of read characters (fixed or variable)

● Search for the expected response

● The first solution can miss a response if delay time is not managed 
properly

● The second solution can be blocking if a shorter (or no) response is 
received 



  

Solution 2

● The implementation of the function, that represents a more flexible 
compromise respect to previous solutions:

uint8_t wait_for_resp(String resp, uint16_t millis_timeout, bool dbg){

  uint8_t retval = 0;  // the return value
  uint8_t str_len = resp.length();  // how many chars are expected in resp
  uint8_t str_i = 0;   // index when comparing
  char in_char;

  unsigned long in_time = millis();  // start time
  volatile unsigned long out_time = in_time + millis_timeout; // quit time

  // ...continues with the loop...

  



  

Solution 2

● ... loop for time-check and read-check...

 

  while(millis() < out_time){    // time-check
    if(Serial.available()){      // non-blocking
      in_char = Serial.read();
      if(dbg) Serial.write(in_char);  // example of debug output
      if(in_char == resp.charAt(str_i)){ // read-check
        str_i++;                         // incrementing index
        out_time = millis() + millis_timeout; // resetting quit time
      } else {
        str_i = 0;  // unexpected char, resetting index
      }
      if(str_i == str_len){  // all chars in response OK 
        retval = 1;  // setting return value 
        break;  // breaking the while, so the function will return immed.
      }
    }
  }

  return retval;  // returning the "response found" or not
}



  

Solution 2

● FYI, during the execution, the function can be called:

  uint8_t r = 0;
  
  //...
  Serial.print("AT+RENEW");
  r = wait_for_resp("OK+RENEW", 100, 0); // wait for 100 milliseconds

  if(r){
    Serial.print("AT+RESET");
    r = wait_for_resp("OK+RESET", 1400, 0);  // wait longer...
  }

  //...
  //...

  if(!r){
    // signal ERROR in some way....
  }

